﻿
&НаСервере
Процедура ПриСозданииНаСервере(Отказ, СтандартнаяОбработка)
	Документ = Параметры.Документ;
	ДатаДокумента = Параметры.ДатаДокумента;
	ТабличныйДокумент = Документы.РасходТовара.ПолучитьМакет("БланкДоставки");
	ТабличныйДокумент.Область("ДатаОформления").Значение = ДатаДокумента;
КонецПроцедуры

&НаКлиенте
Процедура ТабличныйДокументПриИзмененииСодержимогоОбласти(Элемент, Область)
	Если Область.Имя = "ДатаДоставки" Тогда
		ТабличныйДокумент.Область("СрокДоставки").Значение = 
		    Число(ТабличныйДокумент.Область("ДатаДоставки").Значение - НачалоДня(ДатаДокумента)) / (24 * 60 * 60);
	ИначеЕсли Область.Имя = "СрокДоставки" Тогда
		ТабличныйДокумент.Область("ДатаДоставки").Значение = 
		    НачалоДня(ДатаДокумента) + ТабличныйДокумент.Область("СрокДоставки").Значение * 24 * 60 * 60;
	КонецЕсли	
КонецПроцедуры

&НаКлиенте
Процедура ПриОткрытии(Отказ)
	Элементы.ТабличныйДокумент.ТекущаяОбласть = ТабличныйДокумент.Область("Адрес");
КонецПроцедуры

&НаКлиенте
Процедура ПечатьПакетаДокументов(Команда)
	Пакет = ПечатьПакетаДокументовНаСервере();
	Пакет.Напечатать(РежимИспользованияДиалогаПечати.Использовать);
КонецПроцедуры

&НаСервере
Функция ПечатьПакетаДокументовНаСервере()
	Пакет = Новый ПакетОтображаемыхДокументов;
	
	Пакет.Состав.Добавить().Данные = ПоместитьВоВременноеХранилище(ТабличныйДокумент, УникальныйИдентификатор);
	
	РасходнаяНакладная = Новый ТабличныйДокумент;
	Документ.ПолучитьОбъект().ПечатнаяФорма(РасходнаяНакладная);
	Пакет.Состав.Добавить().Данные = ПоместитьВоВременноеХранилище(РасходнаяНакладная, УникальныйИдентификатор);
	
	Возврат Пакет;
КонецФункции
